

Portable Program Review

Vol. 2, No. 4

April, 1986

POLYNOMIAL FITTING

This short curve-fitting program works with as many as 50 sets of X-Y coordinate pairs, and can fit a fifth degree curve. Its author, James Hairston, is a senior system safety engineer with Boeing Corp.

For example, let's find a second-degree curve that would fit the following data:

x	y
0	3.2
1	2.0
2	1.2
3	0.5
4	1.0
5	2.0
6	2.9
7	4.0
8	6.6

The form of the answer will be $Y = AX^2 + BX + C$, and we're looking for the values of A, B and C that will most closely follow the curve described by the eight data points.

The result yielded by the

program is

$$Y = 0.239X^2 - 1.5X + 3.2$$

which has a correlation coefficient, or quality of fit, of 0.992. (A correlation coefficient of 1 or -1 describes a perfect fit; a value of 0 indicates that the data will not at all fit this curve.)

```
1000 CLS
1010 PRINT "This program fits 1st to
      5th degree"
1020 PRINT "polynomials to the input
      data pairs"
1030 PRINT
1040 INPUT "Input the degree desired";
      N
1050 DIM P(6, 13), PX(50), PY(50)
1060 IF N<1 THEN PRINT "Too small"
      :GOTO 1050
1070 IF N>5 THEN PRINT "Too large"
      :GOTO 1050
1080 NI = N + 1
1090 CLS
1100 PRINT "The input data is in X / Y
      pairs"
1110 FOR J = 1 TO 6
1120 FOR K = 1 TO 13
1130 P(J, K) = 0
1140 NEXT K
1150 NEXT J
1160 CLS
      :PRINT
1170 PRINT
1180 BBS = "T"
1190 INPUT "Enter the value of X = ";
      XA
1200 INPUT "Enter the value of Y = ";
      YA
1210 PRINT
1220 PRINT "Last data set bad? Respond
      'N'"
```

IN THIS ISSUE:

POLYNOMIAL FITTING	1
SIMULTANEOUS SOLUTION	3
INEXPENSIVE DATA ACQUISITION	5
THE JOY OF COMPUTING	8


```

1230 PRINT "Was this the final data
      set? Respond `Y'"
1240 PRINT "Enter next data set? Press
      `Enter'"
1250 INPUT BB$
1260 IF BB$ = "N" OR BB$ = "n" THEN
      CLS
      :PRINT
      :PRINT "Reenter data set"
      :GOTO 1170
1270 FOR J = 1 TO (N + 1)
1280 FOR K = 1 TO (N + 1)
1290 P(J, K) = P(J, K) + XA ^ (N + J -
      K)
1300 P(J, K + N + 1) = P(J, K)
1310 NEXT K
1320 YX = YA * XA ^ (J - 1)
1330 P(J, N + N + 3) = P(J, N + N + 3)
      + YX
1340 NEXT J
1350 PX(P(1, NI)) = XA
1360 PY(P(1, NI)) = YA
1370 IF (P(1, NI)) = 50 THEN PRINT
      "Maximum data pairs are 50"
      :GOTO 1390
1380 IF BB$ <> "Y" AND BB$ <> "y" GOTO
      1160
1390 A = 0
1400 B = 0
1410 C = 0
1420 D = 0
1430 E = 0
1440 F = 0
1450 G = 0
1460 CLS
1470 PRINT
1480 PRINT
1490 PRINT
1500 PRINT "                      COMPUTING"
1510 IF N>1 GOTO 1560
1520 G = 1
1530 A = ((P(1, 2) * P(2, 5)) - (P(1,
      1) * P(1, 5))) / ((P(1, 2) * P(2,
      1)) - (P(1, 1) * P(2, 2)))
1540 B = (P(1, 5) - A * P(1, 1)) /
      P(1, 2)
1550 GOTO 2090
1560 K = 0
1570 FOR M = 1 TO (NI + 1)
1580 AJ = 0
1590 FOR L = 1 TO NI
1600 AI = 1
1610 FOR J = 1 TO NI

```

```

1620 K = K + 1
1630 AI = AI * P(J, K)
1640 NEXT J
1650 AJ = AJ + AI
1660 K = L
1670 NEXT L
1680 K = NI + 1
1690 BJ = 0
1700 FOR L = 1 TO NI
1710 BI = 1
1720 FOR J = 1 TO NI
1730 K = K - 1
1740 BI = BI * P(J, K)
1750 NEXT J
1760 BJ = BJ + BI
1770 K = NI + L + 1
1780 NEXT L
1790 K = 0
1800 ON M GOTO 1810, 1840, 1860, 1880,
      1900, 1920, 1940
1810 G = AJ - BJ
1820 IF G = 0 THEN PRINT
      :PRINT "One equation is a multiple
      of another"
      :GOTO 2520
1830 GOTO 1950
1840 A = AJ - BJ
1850 GOTO 1950
1860 B = AJ - BJ
1870 GOTO 1950
1880 C = AJ - BJ
1890 GOTO 1950
1900 D = AJ - BJ
1910 GOTO 1950
1920 E = AJ - BJ
1930 GOTO 1950
1940 F = AJ - BJ
1950 IF M = 1 GOTO 2000
1960 FOR Q = 1 TO NI
1970 P(Q, M - 1) = P(Q, 0)
1980 P(Q, M + NI - 1) = P(Q, 0)
1990 NEXT Q
2000 IF M>NI GOTO 2080
2010 FOR Q = 1 TO NI
2020 P(Q, 0) = P(Q, M)
2030 NEXT Q
2040 FOR Q = 1 TO NI
2050 P(Q, M) = P(Q, NI + NI + 1)
2060 P(Q, M + NI) = P(Q, NI + NI + 1)
2070 NEXT Q
2080 NEXT M
2090 FOR Q = 1 TO NI
2100 ON Q GOTO 2110, 2130, 2150, 2170,

```



```

2190, 2210
2110 CLS
2120 GOTO 2220
2130 PRINT "THE VALUE OF B = "; B / G
2140 GOTO 2220
2150 PRINT "THE VALUE OF C = "; C / G
2160 GOTO 2220
2170 PRINT "THE VALUE OF D = "; D / G
2180 GOTO 2220
2190 PRINT "THE VALUE OF E = "; E / G
2200 GOTO 2220
2210 PRINT "THE VALUE OF F = "; F / G
2220 NEXT Q
2230 BEEP
2240 T1 = 0
2250 T2 = 0
2260 T3 = 0
2270 T4 = 0
2280 T5 = 0
2290 FOR Q = 1 TO (P(1, NI))
2300 ON N GOTO 2310, 2330, 2350, 2370,
2390
2310 YJ = (A * PX(Q) + B) / G
2320 GOTO 2400
2330 YJ = (A * PX(Q) ^ 2 + B * PX(Q) +
C) / G
2340 GOTO 2400
2350 YJ = (A * PX(Q) ^ 3 + B * PX(Q) ^
2 + C * PX(Q) + D) / G
2360 GOTO 2400
2370 YJ = (A * PX(Q) ^ 4 + B * PX(Q) ^
3 + C * PX(Q) ^ 2 + D * PX(Q) +
E) / G
2380 GOTO 2400
2390 YJ = (A * PX(Q) ^ 5 + B * PX(Q) ^
4 + C * PX(Q) ^ 3 + D * PX(Q) ^ 2
+ E * PX(Q) + F) / G
2400 T1 = T1 + (PY(Q) * YJ)
2410 T2 = T2 + YJ
2420 T3 = T3 + PY(Q)
2430 T4 = T4 + YJ ^ 2
2440 T5 = T5 + (PY(Q)) ^ 2
2450 NEXT Q
2460 RI = ((P(1, NI) * T1) - (T2 *
T3)) / ((SQR((P(1, NI) * T4) -
(T2) ^ 2)) * (SQR((P(1, NI) * T5)
- (T3) ^ 2)))
2470 INPUT "Do you want to see the
correlation coefficient?"; EP$
2480 IF EP$ = "Y" OR EP$ = "y" THEN
CLS
:PRINT
:PRINT "The correlation

```

```

coefficient is"; RI
2490 PRINT
2500 INPUT "Do you wish to print the
results?"; EP$
2510 IF EP$ = "Y" OR EP$ = "y" GOTO
2560
2520 INPUT "Do you wish to run this
case again?"; EP$
2530 IF EP$ = "Y" OR EP$ = "y" GOTO
1390
2540 INPUT "Do you wish to run a
different case?"; EP$
2550 IF EP$ = "Y" OR EP$ = "y" GOTO
1000 ELSE MENU
2560 CLS
2570 PRINT
2580 INPUT "When the printer is ready
press Enter; ep$
2590 LPRINT "The form of the equation
is as follows"
2600 ON N GOTO 2610, 2640, 2670, 2700,
2730
2610 LPRINT
2620 LPRINT "AX + B = Y"
2630 GOTO 2750
2640 LPRINT
2650 LPRINT "AX*2 + BX + C = Y"
2660 GOTO 2750
2670 LPRINT
2680 LPRINT "AX*3 + BX*2 + CX + D = Y"
2690 GOTO 2750
2700 LPRINT
2710 LPRINT "AX*4 + BX*3 + CX*2 + DX +
E = Y"
2720 GOTO 2750
2730 LPRINT
2740 LPRINT "AX*5 + BX*4 + CX*3 + DX*2
+ EX + F = Y"
2750 FOR Q = 1 TO NI
2760 ON Q GOTO 2770, 2800, 2830, 2860,
2890, 2920
2770 LPRINT
2780 LPRINT "The value of A = "; A / G
2790 GOTO 2940
2800 LPRINT
2810 LPRINT "The value of B = "; B / G
2820 GOTO 2940
2830 LPRINT
2840 LPRINT "The value of C = "; C / G
2850 GOTO 2940
2860 LPRINT
2870 LPRINT "The value of D = "; D / G
2880 GOTO 2940

```



```

2890 LPRINT
2900 LPRINT "The value of E = "; E / G
2910 GOTO 2940
2920 LPRINT
2930 LPRINT "The value of F = "; F / G
2940 NEXT Q
2950 LPRINT
2960 LPRINT "The correlation
      coefficient = "; RI
2970 LPRINT
2980 GOTO 2520

```

SIMULTANEOUS SOLUTION

To satisfy even the most demanding appetite for number-crunching software, James Hairston has produced SIMEQ.BA, a program which solves simultaneous equations of between two and six unknowns

The solution is obtained by solving an n-by-n matrix of the equations. For example, solving the simultaneous equations

$$\begin{aligned}
 0.25 u - 3.10 v - 0.30 w &= 0.675 \\
 1.25 u + 0.18 v + 0.73 w &= 0.901 \\
 -0.80 u + 1.10 v + 1.60 w &= -2.074
 \end{aligned}$$

yields the following values:

$$\begin{aligned}
 u &= 0.18 \\
 v &= -0.30 \\
 w &= 1.00
 \end{aligned}$$

```

1000 CLS
1010 PRINT "This program solves
      simultaneous"
1020 PRINT "equations of 2 to 6
      unknowns."
1030 PRINT "The solution is obtained
      by solving"
1040 PRINT "an N by N matrix"
1050 PRINT
1060 DIM P(6, 13)
1070 INPUT "Enter the number of
      unknowns"; N
1080 A = 0
1090 B = 0
1100 C = 0
1110 D = 0
1120 E = 0
1130 F = 0

```

```

1140 G = 0
1150 IF N<2 THEN PRINT "Too small"
      :GOTO 1070
1160 IF N>6 THEN PRINT "Too large"
      :GOTO 1070
1170 CLS
1180 PRINT
1190 PRINT "The form of the equation
      is as follows"
1200 PRINT "AU + BV + CW + DX + EY +
      FZ = G"
1210 FOR J = 1 TO N
1220 FOR K = 1 TO N
1230 ON K GOTO 1240, 1260, 1280, 1300,
      1320, 1340
1240 INPUT "Enter A = "; P(J, K)
1250 GOTO 1350
1260 INPUT "Enter B = "; P(J, K)
1270 GOTO 1350
1280 INPUT "Enter C = "; P(J, K)
1290 GOTO 1350
1300 INPUT "Enter D = "; P(J, K)
1310 GOTO 1350
1320 INPUT "Enter E = "; P(J, K)
1330 GOTO 1350
1340 INPUT "Enter F = "; P(J, K)
1350 P(J, K + N) = P(J, K)
1360 NEXT K
1370 INPUT "Enter G = "; P(J, N + N +
      1)
1380 ER$ = ""
1390 INPUT "Enter N to re - enter this
      data, otherwise press Enter"; ER$
1400 IF ER$ = "N" OR ER$ = "n" THEN J
      = J - 1
      :PRINT
      :PRINT "Reenter line"; J + 1
      :PRINT
      :GOTO 1440
1410 PRINT
1420 PRINT "Enter line"; J + 1
1430 PRINT
1440 NEXT J
1450 CLS
1460 PRINT
1470 PRINT
1480 PRINT "                      COMPUTING"
1490 IF N>2 GOTO 1580
1500 A = (P(1, 5) * P(2, 2)) - (P(1,
      2) * P(2, 5))
1510 B = (P(1, 1) * P(2, 5)) - (P(1,
      5) * P(2, 1))
1520 G = (P(1, 1) * P(2, 2)) - (P(1,
      2) * P(2, 1))

```



```

1530 IF G = 0 THEN PRINT
      :PRINT "One equation is a multiple
      of another"
      :GOTO 2310
1540 CLS
1550 PRINT "THE VALUE OF U = "; A / G
1560 PRINT "THE VALUE OF V = "; B / G
1570 GOTO 2260
1580 K = 0
1590 FOR M = 1 TO (N + 1)
1600 AJ = 0
1610 FOR L = 1 TO N
1620 AI = 1
1630 FOR J = 1 TO N
1640 K = K + 1
1650 AI = AI * P(J, K)
1660 NEXT J
1670 AJ = AJ + AI
1680 K = L
1690 NEXT L
1700 K = N + 1
1710 BJ = 0
1720 FOR L = 1 TO N
1730 BI = 1
1740 FOR J = 1 TO N
1750 K = K - 1
1760 BI = BI * P(J, K)
1770 NEXT J
1780 BJ = BJ + BI
1790 K = N + L + 1
1800 NEXT L
1810 K = 0
1820 ON M GOTO 1830, 1860, 1880, 1900,
      1920, 1940, 1960
1830 G = AJ - BJ
1840 IF G = 0 THEN PRINT
      :PRINT "One equation is a multiple
      of another"
      :GOTO 2310
1850 GOTO 1970
1860 A = AJ - BJ
1870 GOTO 1970
1880 B = AJ - BJ
1890 GOTO 1970
1900 C = AJ - BJ
1910 GOTO 1970
1920 D = AJ - BJ
1930 GOTO 1970
1940 E = AJ - BJ
1950 GOTO 1970
1960 F = AJ - BJ
1970 IF M = 1 GOTO 2020
1980 FOR Q = 1 TO N
1990 P(Q, M - 1) = P(Q, 0)

```

```

2000 P(Q, M + N - 1) = P(Q, 0)
2010 NEXT Q
2020 IF M > N GOTO 2100
2030 FOR Q = 1 TO N
2040 P(Q, 0) = P(Q, M)
2050 NEXT Q
2060 FOR Q = 1 TO N
2070 P(Q, M) = P(Q, N + N + 1)
2080 P(Q, M + N) = P(Q, N + N + 1)
2090 NEXT Q
2100 NEXT M
2110 FOR Q = 1 TO N
2120 ON Q GOTO 2130, 2160, 2180, 2200,
      2220, 2240
2130 CLS
2140 PRINT "The value of U = "; A / G
2150 GOTO 2250
2160 PRINT "The value of V = "; B / G
2170 GOTO 2250
2180 PRINT "The value of W = "; C / G
2190 GOTO 2250
2200 PRINT "The value of X = "; D / G
2210 GOTO 2250
2220 PRINT "The value of Y = "; E / G
2230 GOTO 2250
2240 PRINT "The value of Z = "; F / G
2250 NEXT Q
2260 PRINT
2270 INPUT "Do you wish to print the
      results?"; EP$
2280 IF EP$ = "Y" OR EP$ = "y" THEN
      2330
2290 INPUT "Do you wish to run this
      case again?"; EP$
2300 IF EP$ = "Y" OR EP$ = "y" THEN
      1490
2310 INPUT "Do you wish to run a
      different case?"; EP$
2320 IF EP$ = "Y" OR EP$ = "y" THEN
      1070 ELSE MENU
2330 CLS
2340 PRINT
2350 INPUT "Press any key when the
      printer is ready"; EP$
2360 FOR Q = 1 TO N
2370 ON Q GOTO 2380, 2410, 2440, 2470,
      2500, 2530
2380 LPRINT
2390 LPRINT "The value of U = "; A / G
2400 GOTO 2550
2410 LPRINT
2420 LPRINT "The value of V = "; B / G
2430 GOTO 2550
2440 LPRINT

```



```

2450 LPRINT "The value of W = "; C / G
2460 GOTO 2550
2470 LPRINT
2480 LPRINT "The value of X = "; D / G
2490 GOTO 2550
2500 LPRINT
2510 LPRINT "The value of Y = "; E / G
2520 GOTO 2550
2530 LPRINT
2540 LPRINT "The value of Z = "; F / G
2550 NEXT Q
2560 PRINT
2570 GOTO 2290

```

INEXPENSIVE DATA ACQUISITION

By Ken Osborn

Monitoring the environment using a laptop computer is now affordable, thanks to the number of moderately priced analog-digital converters. But although spending \$500 for a 16-channel, 12-bit-accuracy data acquisition device might seem like a steal to some people, I only wanted one channel and about one percent accuracy -- for about \$10.

The application I had in mind involved taking air flow measurements directly from a vane anemometer. Used to measure air flow velocities in chemistry lab fume hoods, this instrument has no electrical output. I intended using a photoemitter/detector combination to provide the necessary transduction. As the blades on the fan spin faster or slower, they'd make and break an optical link between a light-emitting diode (LED) and a photoelectric detector. Electrical pulses from this combination would be sent to pin three of the Model 100 RS-232 port, where they would be counted and converted into air flow measurements.

A Bargain

The diagram shows the pin connections made between a DB-25

male connector and the photoemitter/detector interface. I used an infrared emitter and detector pair available from Radio Shack (catalog number 276-143). Other makes would work as well. Input to the computer from the photodetector is at pins three and 20; pin 20 provides a +5 volt output when the COM port is opened. (Pin 20 is referred to as the "data terminal ready" pin.)

The photoemitter is connected between pins 20 and seven (ground). A 200-ohm resistor in the series protects the LED.

The interface is positioned with the LEDs on opposite sides of the anemometer so that the vanes can pass between them. As the vanes interrupt the light path, the resistance of the photodetector swings from low to high. At each interruption pin three receives a pulse with a peak at about +5 volts.

The serial communications port of the Model 100 transmits bytes to a data buffer at addresses 65350 through 65413. Addresses 65414, 65415 and 65416 are designated as pointers for the buffer. The pointer at address 65414 is incremented each time a pulse received at pin three -- the basis for the RELTYM.BA program.

Getting Results

RELTYM.BA converts the input at pin three to air-velocity measurements. Lines 130-150 initialize variables and open the COM port for input. The number of the fume hood and location within the hood are entered in lines 250 and 270, to make the output more meaningful. This version of RELTYM.BA appends each set of measurements to the RAM file HOOD.DO, but that can be changed to a portable disk drive file or the printer.

Lines 1030-1100 set up a continuous loop that counts the number of pulses. This loop starts

by reading the system time with a call to 6560. The initial time is set to the second. The stop time is then calculated and the program waits at line 1050 until the clock advances. An "M" is displayed to indicate that measurement has started and the buffer pointer is set to zero.

Lines 1070-1090 accumulate the counts, which is arbitrarily reset at a count of 60. The count is printed on each pass through the loop.

When the stop time equals the system time, the program branches to the calculation routine at line 2030. The air-flow rate is calculated from the number of pulse, the elapsed time (eight seconds in this case) and the anemometer calibration factor. The screen pauses for a second; if the reading is acceptable, pressing any key stores the readings into the HOOD.DO data file. If not, the program starts another count.

The calibration factor used to equate pulses with real-world measurements was determined by running a linear regression on a series of raw counts. This process could be adapted to calibrate the program against any similar series of timed pulses.

If you have questions, send me a self-addressed stamped envelope and I'll attempt a response. I can be reached at 1496 Palm Ave., Richmond, CA, 94805.

```

10 ' *****
20 ' *      RELTYM 2.5      *
30 ' *****
50 ' Monitors pulsed input of 1-400
60 ' cycles per second on RS-232 via
70 ' IR detector on pins 20 and 3.
80 ' IR emitter on pins 20 and 7.
100 ' Initialization
120 CLS
    :CLEAR 500
    :MAXFILES = 2
130 ON ERROR GOTO 6000
140 DEFINT A-Z
150 CLOSE

```

```

:OPEN "RAM:HOOD" FOR APPEND AS 1
:OPEN "COM:97I1D" FOR INPUT AS 2
210 ' Hood settings / Title Line
230 CALL 17001
    :PRINT @ 10, " RS-232 Monitor "
    :CALL 17006
240 PRINT @ 100, "";
250 INPUT "Enter hood # "; HN
260 PRINT @ 140, "";
270 INPUT "Position # "; PN
280 PRINT @ 100, " ";
290 PRINT @ 140, " ";
310 ' Screen Layout
330 PRINT @ 87, "      Count =";
340 PRINT @ 171, "Rate =      fpm"
350 PRINT @ 290, "Time = ";
1010 ' Sum pulses / Keep Time
1030 CALL 6560
    :TI = PEEK(63779) ' Initialize time
1040 TS = (TI + 9) MOD 10 ' Stop time
1050 CALL 6560
    :IF TI = PEEK(63779) THEN 1050
        ' Wait for change
1060 PRINT @ 2, "M"
    :POKE 65414, 0
1070 C1 = PEEK(65414)
    :N = N + 1
1080 IF C1 > 60 THEN C2 = C2 + 1
    :POKE 65414, 0
    :C1 = 0
1090 C3 = C1 + C2 * 60
1100 PRINT @ 102, C3;
1110 CALL 6560
    :IF PEEK(63779) <> TS THEN
        1070 ' 8-second interval
2000 ' Calculate flow / Reset Counters
2020 POKE 65414, 0
    :C2 = 0
    :PRINT @ 2, " ";
2030 R! = 1.23 * C3
2040 PRINT @ 178, USING "###", INT(R!)
2050 PRINT @ 300, TIMES$
2060 FOR WT = 1 TO 200
    :PRINT @ 285, "OK?";
    :NEXT
2070 PRINT @ 285, " ";
2080 IF PEEK(65450) > 0 THEN PRINT
    #1, " "; HN; " ";
    PN; " "; R
    :POKE 65450, 0
    :GOTO 120
5700 GOTO 1030
6000 RESUME 100

```

THE JOY OF COMPUTING

The Model 100 isn't just a great portable notebook computer. Its a real computers, akin to larger, more specialized devices, including industrial minis and mainframes.

Our favorite laptops are among the most inexpensive -- and most flexible -- complete computer systems available, with the BASIC programming language, built-in display, battery power, modem, and serial and parallel ports.

As the articles and programs in this issue demonstrate, the Model 100 and Tandy 200 can compete in the games normally reserved for multi-thousand dollar devices: number crunching and real-world interfacing.

Not bad for a portable notebook computer.

Alan L. Zeichick
Editor

PORTABLE PROGRAM REVIEW
P.O. Box 250
Highland Mill
Camden, ME 04843

FIRST CLASS U.S. POSTAGE PAID CAMDEN, ME PERMIT NO. 5

SUBSCRIBER: PLEASE ADVISE IF ABOVE ADDRESS IS INCORRECT

PORTABLE PROGRAM REVIEW is published monthly and copyright © 1986, all rights reserved, by Camden Communications Inc., a Maine corporation with offices at Highland Mill (P.O. Box 250), Camden, Maine 04843. Contents of this publication may not be reproduced in whole or in part unless expressly authorized in writing by the publisher. Tandy and Radio Shack are registered trademarks of Tandy Corporation. U.S. Subscription \$29.97. Please address all subscriptions and inquiries to Nancy Wight, Circulation Director, (207) 236-4365. ISSN: 0883-7295.